

## F07FWF (CPOTRI/ZPOTRI) – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F07FWF (CPOTRI/ZPOTRI) computes the inverse of a complex Hermitian positive-definite matrix  $A$ , where  $A$  has been factorized by F07FRF (CPOTRF/ZPOTRF).

### 2 Specification

```
SUBROUTINE F07FWF(UPLO, N, A, LDA, INFO)
ENTRY      cpotri(UPLO, N, A, LDA, INFO)
INTEGER    N, LDA, INFO
complex  A(LDA,*)
CHARACTER*1 UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

To compute the inverse of a complex Hermitian positive-definite matrix  $A$ , the routine must be preceded by a call to F07FRF (CPOTRF/ZPOTRF), which computes the Cholesky factorization of  $A$ .

If UPLO = 'U',  $A = U^H U$  and  $A^{-1}$  is computed by first inverting  $U$  and then forming  $(U^{-1})(U^{-1})^H$ .

If UPLO = 'L',  $A = LL^H$  and  $A^{-1}$  is computed by first inverting  $L$  and then forming  $(L^{-1})^H(L^{-1})$ .

### 4 References

- [1] Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

### 5 Parameters

- 1: UPLO — CHARACTER\*1 *Input*  
*On entry:* indicates whether  $A$  has been factorized as  $U^H U$  or  $LL^H$  as follows:  
     if UPLO = 'U', then  $A = U^H U$ , where  $U$  is upper triangular;  
     if UPLO = 'L', then  $A = LL^H$ , where  $L$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N — INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: A(LDA,\*) — *complex* array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the upper triangular matrix  $U$  if UPLO = 'U' or the lower triangular matrix  $L$  if UPLO = 'L', as returned by F07FRF (CPOTRF/ZPOTRF).  
*On exit:*  $U$  is overwritten by the upper triangle of  $A^{-1}$  if UPLO = 'U';  $L$  is overwritten by the lower triangle of  $A^{-1}$  if UPLO = 'L'.

4: LDA — INTEGER *Input*

*On entry:* the first dimension of the array A as declared in the (sub)program from which F07FWF (CPOTRI/ZPOTRI) is called.

*Constraint:* LDA  $\geq$  max(1,N).

5: INFO — INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the  $i$ th diagonal element of the Cholesky factor is zero; the Cholesky factor is singular and the inverse of  $A$  cannot be computed.

## 7 Accuracy

The computed inverse  $X$  satisfies

$$\|XA - I\|_2 \leq c(n)\epsilon\kappa_2(A) \quad \text{and} \quad \|AX - I\|_2 \leq c(n)\epsilon\kappa_2(A),$$

where  $c(n)$  is a modest function of  $n$ ,  $\epsilon$  is the *machine precision* and  $\kappa_2(A)$  is the condition number of  $A$  defined by

$$\kappa_2(A) = \|A\|_2 \|A^{-1}\|_2.$$

## 8 Further Comments

The total number of real floating-point operations is approximately  $\frac{8}{3}n^3$ .

The real analogue of this routine is F07FJF (SPOTRI/DPOTRI).

## 9 Example

To compute the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} 3.23 + 0.00i & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 + 0.00i & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 + 0.00i & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 + 0.00i \end{pmatrix}.$$

Here  $A$  is Hermitian positive-definite and must first be factorized by F07FRF (CPOTRF/ZPOTRF).

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*   F07FWF Example Program Text
*   Mark 16 Release. NAG Copyright 1993.
*   .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA
PARAMETER       (NMAX=8,LDA=NMAX)
*   .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*   .. Local Arrays ..
complex        A(LDA,NMAX)
CHARACTER       CLABS(1), RLABS(1)
*   .. External Subroutines ..
EXTERNAL        X04DBF, cpotrf, cpotri
*   .. Executable Statements ..
WRITE (NOUT,*) 'F07FWF Example Program Results'
*   Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*       Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF
*
*       Factorize A
*
      CALL cpotrf(UPLO,N,A,LDA,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*         Compute inverse of A
*
        CALL cpotri(UPLO,N,A,LDA,INFO)
*
*         Print inverse
*
        IFAIL = 0
        CALL X04DBF(UPLO,'Nonunit',N,N,A,LDA,'Bracketed','F7.4',
+              'Inverse','Integer',RLABS,'Integer',CLABS,80,0,
+              IFAIL)
      ELSE
        WRITE (NOUT,*) 'A is not positive-definite'
      END IF
    END IF
  STOP
*
  END

```

## 9.2 Program Data

F07FWF Example Program Data

```
4                                     :Value of N
'L'                                   :Value of UPL0
(3.23, 0.00)
(1.51, 1.92) ( 3.58, 0.00)
(1.90,-0.84) (-0.23,-1.11) ( 4.09, 0.00)
(0.42,-2.50) (-1.18,-1.37) ( 2.33, 0.14) ( 4.29, 0.00) :End of matrix A
```

## 9.3 Program Results

F07FWF Example Program Results

Inverse

```
                                     1           2           3           4
1 ( 5.4691, 0.0000)
2 (-1.2624,-1.5491) ( 1.1024, 0.0000)
3 (-2.9746,-0.9616) ( 0.8989,-0.5672) ( 2.1589, 0.0000)
4 ( 1.1962, 2.9772) (-0.9826,-0.2566) (-1.3756,-1.4550) ( 2.2934, 0.0000)
```

---